
The Calculated Typer
(Functional Pearl)

Zac Garby
University of Nottingham

United Kingdom
zac.garby@nottingham.ac.uk

Patrick Bahr
IT University of Copenhagen

Denmark
paba@itu.dk

Graham Hutton
University of Nottingham

United Kingdom
graham.hutton@nottingham.ac.uk

Abstract
We present a calculational approach to the design of type
checkers, showing how they can be derived from behavioural
specifications using equational reasoning. We focus on lan-
guages whose semantics can be expressed as a fold, and show
how the calculations can be simplified using fold fusion. This
approach enables the compositional derivation of correct-by-
construction type checkers based on solving and composing
fusion preconditions. We introduce our approach using a
simple expression language, to which we then add support
for exception handling and checked exceptions.

CCSConcepts: • Software and its engineering→ Formal
software verification; •Theory of computation→ Logic
and verification; Type theory; Program verification.

1 Introduction
Type checking ensures that programs are constructed in
accordance with a set of typing rules [14]. For example, if
we specify that addition requires two integer arguments and
produces an integer result, then the expression 1 + 2 is well
typed because both arguments are integers, whereas 1+True
is ill typed because the second argument is not an integer.
Moreover, we can say that 1 + 2 has integer type because it
produces an integer result, while 1 + True has a type error
due to its improper construction.

In this article, we focus on the problem of defining type-
checking functions that analyse the structure of a program
to determine whether it is well typed. If the program is well
typed, the type-checking function returns the type of the
program, and otherwise it returns a type error. Tradition-
ally, such type checkers are defined by hand based on a
predetermined set of typing rules that specify valid ways
to construct programs. Here we take a different approach,
showing how to calculate type-checking functions directly
from specifications of their behaviour.
The starting point for our calculational approach is a se-

mantics for the language being considered, expressed as an
evaluation function. We then formulate a specification that
captures the desired behaviour of the type-checking func-
tion, which here means that it returns the type of the value
that would be produced if evaluation succeeds, or returns
a type error if evaluation may fail due to the input being
badly formed in some way. Finally, we use equational rea-
soning techniques to calculate an implementation for the
type-checking function that satisfies the specification.
The calculational approach to type checker design has a

number of benefits. First of all, the resulting type checkers
are correct-by-construction [1], eliminating the need for sepa-
rate correctness proofs. Secondly, the approach provides a
systematic way to discover typing rules, and to explore al-
ternative design choices during the calculation process. And
finally, it is readily amenable to mechanical formalisation,
enabling the use of proof assistants as interactive tools for
developing and certifying the calculations.
Our goal is not to devise a general-purpose approach to

designing type systems for full-featured programming lan-
guages. Instead, we explore the underlying algebraic struc-
ture of type checking, and exploit this structure to enable the
derivation of correct-by-construction type checkers using
simple, compositional reasoning principles. To this end, we
focus on terminating, first-order languages.

We develop our approach in three stages. We begin with a
first-principles approach using explicit recursive definitions
and inductive reasoning. We then simplify the calculations
by adopting an algebraic approach using a fold operator
and its fusion property. Finally, we refine the methodology
further by using a constraint-based approach to solving and
composing fusion preconditions. We introduce our approach
using a simple expression language with conditionals, then

extend the language with support for exception handling,
and finally consider checked exceptions.

All our programs are written in Haskell [10], but the ideas
are applicable in any functional programming language. Be-
cause the calculations are the central focus, they are typically
presented in detail rather than being compressed or omitted.
Haskell code for each of the examples, together with an Agda
formalisation of all of the main results, is freely available
online as supplementary material [6].

2 Positivity Checking
Type checking is an example of static analysis [13], which
seeks to reason about the behaviour of programs without
executing them. Prior to considering type checking, we first
consider a simpler example, to illustrate how a static analysis
can be calculated from a specification of its behaviour using
equational reasoning techniques.

Consider a simple type of arithmetic expressions built up
from integer values using an addition operator, together with
a semantics that evaluates an expression to an integer:

data Expr = Val Int | Add Expr Expr

eval :: Expr → Int
eval (Val n) = n
eval (Add x y) = eval x + eval y

Suppose now that wewish to define a function isPos::Expr →
Bool that decides if an expression is positive, without evalu-
ating the expression. The desired behaviour is as follows:

isPos e ⇒ eval e > 0

That is, if an expression is classified as positive, then evalua-
tion should give a positive result. Note that the specification
is an implication rather than an equivalence, because in
general it may not be possible to decide if an expression is
positive without evaluating it.

To calculate a definition for isPos, we proceed by induction
on the expression e. In each case, we start with the right-hand
side of the specification, eval e>0, and seek to strengthen it to
something of the form isPos e by defining a suitable clause for
the function in this case. Each step in the calculation is either
a logical equivalence (⇔) or a reverse implication (⇐), so
that the final term of the calculation, isPos e, implies the
term that we started with, eval e > 0.

Case: Val n

eval (Val n) > 0
⇔ { applying eval }
n > 0

⇔ { define: isPos (Val n) = n > 0 }
isPos (Val n)

Case: Add x y

eval (Add x y) > 0
⇔ { applying eval }
eval x + eval y > 0

⇐ { addition preserves positivity }
eval x > 0 ∧ eval y > 0

⇐ { induction hypotheses }
isPos x ∧ isPos y

⇔ { define: isPos (Add x y) = isPos x ∧ isPos y }
isPos (Add x y)

The key step in the calculation uses the fact that addition
preserves positivity, i.e. n > 0 and m > 0 implies n +m > 0, to
transform the term being manipulated into a form to which
the induction hypotheses can be applied. In conclusion, we
have calculated the following definition:

isPos :: Expr → Bool
isPos (Val n) = n > 0
isPos (Add x y) = isPos x ∧ isPos y

The approach we use to calculate type checking functions is
similar to the above, but in a more sophisticated setting.

3 Conditional Expressions
We now consider an expression language built up from basic
values using addition and conditional operations, where a
value is either an integer, a logical value, or an error value:

data Expr = Val Value | Add Expr Expr | If Expr Expr Expr

data Value = I Int | B Bool | Error
The semantics is defined using a function that evaluates an
expression, with the error value being returned if evaluation
fails due to the expression being badly formed:

eval :: Expr → Value
eval (Val v) = v
eval (Add x y) = add (eval x) (eval y)
eval (If x y z) = cond (eval x) (eval y) (eval z)
The operations add and cond on values state that addition
requires two integers to succeed, while conditionals require
a logical value to make a choice between two alternatives:

add :: Value → Value → Value
add (I n) (I m) = I (n +m)
add = Error

cond :: Value → Value → Value → Value
cond (B b) v w = if b then v else w
cond = Error

As we shall see, separating out these operations as named
functions plays an important role in our development, as it
allows us to exploit properties of these functions.

Badly-formed expressions could also be handled in other
ways, such as by replacing the explicit error value by the
Maybe monad and defining eval :: Expr → Maybe Value, or

by using a big-step operational approach and defining an
evaluation relation between expressions and values, which
doesn’t require an error value as relations can be partial. Here
we choose to focus on the simple functional approach with
an explicit error value, as this mirrors the first-principles
approach used by Bahr and Hutton [2].

4 Type Checking
The evaluation function eval fails if an expression is badly
formed. In this case, badly formed means that it contains
a type error, such as attempting to add two values that are
not integers. To formalise this idea, we first define a simple
language of types, comprising integers, logical values, and
an error type, together with a function that abstracts from
the contents of a value and returns its type:

data Type = INT | BOOL | ERROR
tval :: Value → Type
tval (I ) = INT
tval (B ) = BOOL
tval Error = ERROR

Suppose now that we wish to define a type checking function
texp :: Expr → Type that returns the type of an expression,
with the error type being returned if the expression is badly
formed. How can we specify the desired behaviour of this
function? A first attempt might be as follows:

texp e = tval (eval e)

This equation states that type checking an expression should
give the same result as evaluating the expression and taking
the type of the resulting value. While this captures the basic
idea, it is also too strong. To see why, consider the following
expression, written in regular Haskell syntax:

if True then 1 else False

Evaluating this expression gives the value 1, an integer. How-
ever, for the type checker to determine this, it would need
to use the value of the condition to decide which branch
applies. We do not want the type checker to observe values
in this way, as in general this involves evaluation, which is
precisely what the type checker aims to abstract away.
The above specification would therefore be unsatisfiable

under this constraint. The root cause is that type checking
is necessarily approximate rather than precise. We can for-
malise this idea by defining an ordering relation on types
via the following instance declaration:

instance Ord Type where
(⩽) :: Type → Type → Bool
t ⩽ t′ = t == ERROR ∨ t == t′

That is, ERROR is defined to be less than any type, while any
two other types are related by the ordering if they are equal.
As a Hasse diagram, the ordering can be illustrated as:

INT BOOL

ERROR

We can view this as an information ordering, where t ⩽ t′

means the type t contains less information than the type t′.
This is the reverse of the usual subtype ordering, and is used
to capture the approximate nature of type checking. We will
sometimes find it convenient to use the opposite ordering ⩾,
defined in the usual way by t ⩾ t′ iff t′ ⩽ t.

Note that the Ord class normally requires a total ordering,
but for our purposes we only assume the ⩽ ordering is par-
tial and do not use properties or operations that depend on
totality. Using this ordering, the desired behaviour of type
checking can then be captured as follows:

texp e ⩽ tval (eval e)

That is, type checking either returns a type error, or gives
the same result as evaluating the given expression and then
taking the type of the resulting value.
Note that the above specification gives flexibility in how

type checking is implemented, as theremay bemany possible
definitions that satisfy the inequation, with some being more
informative than others. Indeed, the trivial definition texp e =
ERROR that always gives a type error is perfectly valid. Our
calculational approach naturally avoids such trivial solutions,
and provides a systematic means for designing type checkers
that satisfy the above specification.

5 Calculating the Type Checker
Rather than first defining the function texp and then proving
it is correct, we can also use the specification

texp e ⩽ tval (eval e)

as the basis for calculating the definition of texp. The calcula-
tion proceeds by induction on the expression e. For each case
of e, we start with the right-hand side of the specification,
tval (eval e), and seek to strengthen it into something of the
form texp e by defining a suitable clause for the function in
this case. Generalising from Section 2, each calculation step
is either an equality (=) or a reverse inclusion (⩾).

Case: Val v

tval (eval (Val v))
= { applying eval }
tval v

= { define: texp (Val v) = tval v }
texp (Val v)

The last step introduces a clause for texp so that the final
term of the calculation is of the desired form that matches
the left-hand side of the specification.

Case: Add x y

tval (eval (Add x y))
= { applying eval }
tval (add (eval x) (eval y))

Now we appear to be stuck, as no further definitions can
be applied. However, as we are performing an inductive
calculation, we can use the induction hypotheses:

texp x ⩽ tval (eval x)
texp y ⩽ tval (eval y)

To use these hypotheses, it is clear that we need to dis-
tribute tval over the operation add on values in the term
tval (add (eval x) (eval y)) to give a term of the form
add′ (tval (eval x)) (tval (eval y)), for some operation
add′ on types. That is, we need to solve the inequation:

tval (add (eval x) (eval y))
⩾

add′ (tval (eval x)) (tval (eval y))
First of all, we generalise from the specific values eval x and
eval y to arbitrary values:

tval (add v w) ⩾ add′ (tval v) (tval w)
Assuming this property then allows us to apply induction
and complete the calculation:

tval (add (eval x) (eval y))
⩾ { assume: tval (add v w) ⩾ add′ (tval v) (tval w) }
add′ (tval (eval x)) (tval (eval y))
⩾ { induction hypotheses, assume: add′ is monotonic }
add′ (texp x) (texp y)

= { define: texp (Add x y) = add′ (texp x) (texp y) }
texp (Add x y)

The induction step also requires that add′ is monotonic, i.e.
preserves the ordering on types:

t ⩽ t′ u ⩽ u′

add′ t u ⩽ add′ t′ u′

To discharge the assumptions made in the above calculation,
it remains to define the operation add′ :: Type → Type →
Type on types, and show that it satisfies the required dis-
tributivity and monotonicity properties. In fact, we can cal-
culate the definition for add′ directly from its distributivity
property by starting with the left-hand side of this property,
tval (add v w), and aiming to transform it into the form of
the right-hand side, add′ (tval v) (tval w):

tval (add v w)
= { applying add }
tval (case (v,w) of

(I n, I m) → I (n +m)
( , ) → Error)

= { distribution, tval is strict }
case (v,w) of

(I n, I m) → tval (I (n +m))
( , ) → tval Error

= { applying tval }
case (v,w) of

(I n, I m) → INT
( , ) → ERROR

= { unapplying tval }
case (tval v, tval w) of

(INT , INT ) → INT
( , ) → ERROR

= { define: add′ INT INT = INT ; add′ = ERROR }
add′ (tval v) (tval w)

In summary, we have established an equality between the
two sides, which is stronger than the required inequality,
and have therefore calculated the following definition:

add′ :: Type → Type → Type
add′ INT INT = INT
add′ = ERROR

That is, adding two integers gives an integer, while adding
anything else gives a type error. We can show add′ is mono-
tonic by considering monotonicity in each argument sepa-
rately. For example, we can show monotonicity in the first
argument by the following reasoning:

add′ t u ⩽ add′ t′ u
⇔ { applying ⩽ }
add′ t u == ERROR ∨ add′ t u == add′ t′ u

⇐ { unapplying add′ }
t == ERROR ∨ add′ t u == add′ t′ u

⇐ { substitution }
t == ERROR ∨ t == t′

⇔ { unapplying ⩽ }
t ⩽ t′

The key step here is the second one, where we use the fact
that add′ returns an error if its first argument is an error.
Monotonicity in the second argument follows similarly.

Case: If x y z
This case proceeds in a similar manner to the case for addi-
tion, by assuming a distributivity property that allows the
induction hypotheses to be applied:

tval (eval (If x y z))
= { applying eval }
tval (cond (eval x) (eval y) (eval z))
⩾ {assume: tval (cond c v w) ⩾

cond′ (tval c) (tval v) (tval w) }
cond′ (tval (eval x)) (tval (eval y)) (tval (eval z))
⩾ { induction hypotheses, assume: cond′ is monotonic }
cond′ (texp x) (texp y) (texp z)

= {define: texp (If x y z) =

cond′ (texp x) (texp y) (texp z) }
texp (If x y z)

We can then calculate the definition for cond′ from its dis-
tributivity property by starting with the left-hand side, and
aiming for the right-hand side:

tval (cond c v w)
= { applying cond }
tval (case c of
B b → if b then v else w

→ Error)
= { distribution, tval is strict }
case c of
B b → if b then tval v else tval w

→ tval Error
= { applying tval }
case c of
B b → if b then tval v else tval w

→ ERROR
⩾ { lemma below, case is monotonic }
case c of
B b → if tval v == tval w then tval v else ERROR

→ ERROR
= { unapplying tval }
case (tval c) of
BOOL → if tval v == tval w then tval v else ERROR

→ ERROR
= {define:

cond′ BOOL t t′ = if t == t′ then t else ERROR
cond′ = ERROR }

cond′ (tval c) (tval v) (tval w)

The key step here is a lemma that allows us to replace a
conditional that depends on a logical value by a conditional
that only depends on the types in the branches:

if b then t else t′ ⩾ if t == t′ then t else ERROR

We can prove this lemma by case analysis on t == t′. If this
value is true, the lemma simplifies to (if b then t else t) ⩾ t,
which in turn simplifies to t ⩾ t assuming b is well-defined,
which is then true by reflexivity. If t == t′ is false, the lemma
simplifies to (if b then t else t′) ⩾ ERROR, which is true
because ERROR is the smallest type by definition.
In summary, we have calculated the following definition

for cond′, which formalises the idea that a conditional ex-
pression requires a logical value and two branches that have
the same type, otherwise it gives a type error:

cond′ :: Type → Type → Type → Type
cond′ BOOL t t′ = if t == t′ then t else ERROR
cond′ = ERROR

Showing that cond′ is monotonic on types proceeds in a
similar manner to the function add′, using the fact that cond′
preserves type errors in all arguments.

Putting everything together, we have calculated the follow-
ing definition for the type-checking function texp in terms
of newly defined operations add′ and cond′ on types:

texp :: Expr → Type
texp (Val v) = tval v
texp (Add x y) = add′ (texp x) (texp y)
texp (If x y z) = cond′ (texp x) (texp y) (texp z)
add′ :: Type → Type → Type
add′ INT INT = INT
add′ = ERROR

cond′ :: Type → Type → Type → Type
cond′ BOOL t t′ = if t == t′ then t else ERROR
cond′ = ERROR

6 Using Fold and Fusion
An important aspect of our development above was the use
of operations add and cond on values, and operations add′
and cond′ on types. In particular, separating these out as
named functions allowed us to state and exploit algebraic
properties of these operations, namely distributivity and
monotonicity, to simplify and guide our calculations.

However, we can take the algebraic approach further, and
there are benefits from doing so. The starting point is to
define a fold operator for expressions [12], which takes the
operation to apply for each form of expression as an addi-
tional parameter in the definition:

folde :: (Value → a) → (a → a → a) →
(a → a → a → a) → Expr → a

folde val add cond = f where
f (Val v) = val v
f (Add x y) = add (f x) (f y)
f (If x y z) = cond (f x) (f y) (f z)

Using this operator, we can then define evaluation and type
checking in a more concise manner by supplying the appro-
priate operations for values, addition and conditionals:

eval :: Expr → Value
eval = folde id add cond

texp :: Expr → Type
texp = folde tval add′ cond′

The fold operator also has a useful fusion property [12],
which allows a function h that satisfies suitable distribu-
tivity, or homomorphism, properties to be fused with a fold
over one collection of operations, val, add and cond, to give
a fold over another collection, val′, add′ and cond′. In our
setting, we use a variant of fusion where equality (=) is re-
placed by a pre-order (⩾), under which the operations add′

and cond′ are required to be monotonic:

h (val x) ⩾ val′ x
h (add x y) ⩾ add′ (h x) (h y)

h (cond x y z) ⩾ cond′ (h x) (h y) (h z)
h (folde val add cond e) ⩾ folde val′ add′ cond′ e

This fusion property can be proved by straightforward induc-
tion on expressions. Using the above ideas, it now becomes
evident that the correctness of type checking is just an ap-
plication of fold fusion. In particular, the specification

tval (eval e) ⩾ texp e

can now be expanded to give:

tval (folde id add cond e) ⩾ folde tval add′ cond′ e

Hence, by fusion, this property is true if the operations add′
and cond′ are monotonic, and the function tval satisfies the
following homomorphism properties:

tval (id x) ⩾ tval x

tval (add x y) ⩾ add′ (tval x) (tval y)
tval (cond x y z) ⩾ cond′ (tval x) (tval y) (tval z)

The first property is trivially true by reflexivity, while the
second and third lead to the same calculations and definitions
for add′ and cond′ as shown in the previous section.

In conclusion, using an algebraic approach allows the type-
checking function texp :: Expr → Type to be obtained in a
more principled manner than previously. In particular, using
the fold operation and its associated fusion property makes
explicit the key property required for the type checking
function to be correct, namely that tval :: Value → Type is
a homomorphism between basic operations on values (add
and cond) and those on types (add′ and cond′).

7 Using Constraints
As shown in the previous section, the correctness of type
checking depends on certain homomorphism properties. For
example, for conditionals we must establish that:

tval (cond x y z) ⩾ cond′ (tval x) (tval y) (tval z)

Earlier we showed how this property can be used as the
basis for calculating the definition for the operation cond′ on
types. However, an unsatisfactory aspect of the calculation
was that it required ‘inventing’ the key lemma that makes
the calculation work, namely:

if b then t else t′ ⩾ if t == t′ then t else ERROR

However, there is another approach to calculating cond′ that
is both simpler and avoids the need to invent a lemma. The
approach is based on using the homomorphism property that
specifies the desired behaviour of cond′ to derive a number
of simpler constraints, which can then be solved collectively
to give a definition for cond′ itself.

We begin by observing that the operation cond on values
is defined by case analysis on its first argument. We then pro-
ceed by considering the four possible cases of this argument,
and simplifying the homomorphism property for condition-
als in each of these cases. In each case, we seek to obtain a
simpler property that implies the homomorphism property.
Hence, as in Section 2, each calculation step is either a logical
equivalence or a reverse implication.

Case: x = B True

tval (cond (B True) y z) ⩾
cond′ (tval (B True)) (tval y) (tval z)

⇔ { applying cond, tval }
tval y ⩾ cond′ BOOL (tval y) (tval z)

⇐ { generalising to arbitrary types }
t ⩾ cond′ BOOL t t′

Case: x = B False

tval (cond (B False) y z) ⩾
cond′ (tval (B False)) (tval y) (tval z)

⇔ { applying cond, tval }
tval z ⩾ cond′ BOOL (tval y) (tval z)

⇐ { generalising to arbitrary types }
t′ ⩾ cond′ BOOL t t′

Case: x = I n

tval (cond (I n) y z) ⩾
cond′ (tval (I n)) (tval y) (tval z)

⇔ { applying cond, tval }
tval Error ⩾ cond′ INT (tval y) (tval z)

⇔ { applying tval }
ERROR ⩾ cond′ INT (tval y) (tval z)

⇐ { generalising to arbitrary types }
ERROR ⩾ cond′ INT t t′

Case: x = Error

tval (cond Error y z) ⩾
cond′ (tval Error) (tval y) (tval z)

⇔ { applying cond, tval }
tval Error ⩾ cond′ ERROR (tval y) (tval z)

⇔ { applying tval }
ERROR ⩾ cond′ ERROR (tval y) (tval z)

⇐ { generalising to arbitrary types }
ERROR ⩾ cond′ ERROR t t′

The above reasoning shows that the following four con-
straints for the operation cond′ are together sufficient to

satisfy the homomorphism property for conditionals:
cond′ BOOL t t′ ⩽ t

cond′ BOOL t t′ ⩽ t′

cond′ INT t t′ ⩽ ERROR

cond′ ERROR t t′ ⩽ ERROR

The last two constraints have a unique solution, given by
defining cond′ INT t t′ = ERROR and cond′ ERROR t t′ =
ERROR, because ERROR is the smallest type. In turn, if we
assume a greatest lower bound operator ⊓ on types, then
the first two constraints are together equivalent to

cond′ BOOL t t′ ⩽ t ⊓ t′

by the universal property of greatest lower bounds:
x ⩽ y and x ⩽ z iff x ⩽ y ⊓ z

Hence, we can define cond′ BOOL t t′ = t ⊓ t′ as the
optimal, i.e. greatest or most informative, solution to the first
two constraints. Putting all of the above reasoning together,
we conclude that the cond′ operation on types can be defined
by the following three equations,
cond′ BOOL t t′ = t ⊓ t′

cond′ INT t t′ = ERROR
cond′ ERROR t t′ = ERROR

which can then be simplified by using the wildcard pattern
to combine the last two cases:
cond′ BOOL t t′ = t ⊓ t′

cond′ = ERROR

For the ordering relation ⩽ on types that we are using, where
t ⩽ t′ iff t = ERROR or t = t′, the greatest lower bound
operator is defined simply as:
(⊓) :: Type → Type → Type
t ⊓ t′ = if t == t′ then t else ERROR

In summary, we have shown how the previous definition
for the cond′ operation can be calculated in a principled
manner that does not require the invention of a lemma:
cond′ :: Type → Type → Type → Type
cond′ BOOL t t′ = if t == t′ then t else ERROR
cond′ = ERROR

The same kind of constraint-based reasoning can also be
used to calculate the operation add′ on types from the ho-
momorphism property that it must satisfy:

tval (add x y) ⩾ add′ (tval x) (tval y)
The operation add on values is defined by case analysis on
its two arguments, so we proceed by considering the two
cases for the definition separately.

Case: x = I n and y = I m

tval (add (I n) (I m)) ⩾ add′ (tval (I n)) (tval (I m))
⇔ { applying add, tval }

tval (I (n +m)) ⩾ add′ INT INT
⇔ { applying tval }
INT ⩾ add′ INT INT

Case: x ≠ I n or y ≠ I m

tval (add x y) ⩾ add′ (tval x) (tval y)
⇔ { applying add }
tval Error ⩾ add′ (tval x) (tval y)

⇔ { applying tval }
ERROR ⩾ add′ (tval x) (tval y)

⇐ { generalising to arbitrary types }
ERROR ⩾ add t t′, if t ≠ INT or t′ ≠ INT

The above reasoning shows that the following two con-
straints for the operation add′ are together sufficient to sat-
isfy the homomorphism property for addition:

add′ INT INT ⩽ INT

add′ t t′ ⩽ ERROR if t ≠ INT or t′ ≠ INT

The optimal, i.e. greatest or most informative, solution to
these inequations is to strengthen each to an equality by
defining add′ INT INT = INT and add′ t t′ = ERROR if
t ≠ INT or t′ ≠ INT , which can then be simplified using the
wildcard pattern to give the following definition:

add′ :: Type → Type → Type
add′ INT INT = INT
add′ = ERROR

In conclusion, adopting a constraint-based approach to
solving the required homomorphism properties allows the
basic type checking operations add′ and cond′ to be obtained
in a simpler way than previously, and in a manner that does
not require inventing a lemma.

8 Composing Constraints
In this section we improve the constraint-based approach
from the previous section, by showing how type checking
operations can be defined by directly composing their con-
straints. This approach also ensures that the resulting opera-
tions are monotonic by construction, and gives flexibility to
handle more complex types in later sections.
We introduce and explain the compositional approach

by considering the first constraint that we derived for the
operation cond′ on types in the previous section:

cond′ BOOL t t′ ⩽ t (1)

We will transform this constraint into an equivalent form
that can be used to compose the definition for cond′ directly.
We proceed in a series of steps. First of all, we observe that
under the assumption that cond′ is monotonic, the above
constraint is equivalent to:

cond′ s t t′ ⩽ t if s ⩽ BOOL (2)

That is, the explicit matching on BOOL in the original con-
straint is replaced by the side-condition that the type of the
first argument is bounded above by BOOL. To verify that (2)
implies (1), we simply take s = BOOL. The converse implica-
tion from (1) to (2) can be verified as follows:

cond′ s t t′

⩽ { monotonicity of cond′, assumption s ⩽ BOOL }
cond′ BOOL t t′

⩽ { assumption (1) }
t

Aswe shall see, formulating the constraint as an upper bound
on the first argument rather than an equality will ensure
that the derived definition for cond′ is monotonic.
In the next step, we transform (2) to remove the need

for s ⩽ BOOL as a side condition, by integrating it into the
constraint itself. To achieve this, we extend the language of
types with a greatest element, which we write as TOP :

TOP

INT BOOL

ERROR

The new element TOP is only needed for the purpose of calcu-
lation. Indeed, from our specification texp e ⩽ tval (eval e),
it follows that TOP can never arise as the type of an expres-
sion, because it is not in the image of the function tval that
returns the type of values, and hence we can never have an
expression for which texp e = TOP as this would violate
the specification. Using this new element, we can now write
constraint (2) in the following equivalent form:

cond′ s t t′ ⩽ if s ⩽ BOOL then t else TOP (3)

In this manner, the side condition s ⩽ BOOL is now inte-
grated into the constraint itself. It is easy to see that (3) is
equivalent to (2), because whenever s ⩽ BOOL is false, it
simplifies to cond′ s t t′ ⩽ TOP , which is true by defini-
tion. This form of constraint will be used often, so we find it
convenient to introduce some special notation:

(⇛ ) :: Bool → Type → Type
b ⇛ t = if b then t else TOP

Using this notation, we can then write (3) simply as:

cond′ s t t′ ⩽ (s ⩽ BOOL) ⇛ t (4)

This completes the transformation process. Constraints pro-
duced by following the above steps may at first sight seem
rather unusual, but have a natural operational reading. For
example, the derived constraint (4) can be read as “if the
first argument of a conditional could have type BOOL, then
the conditional as a whole could have the type of the first
branch.” Note that the use of could have here rather than has
reflects the use of inclusion ⩽ rather than equality =.

We can apply the same transformation steps to the other
constraints on cond′ that were calculated in Section 7 to give
the following equivalent set of constraints:

cond′ s t t′ ⩽ (s ⩽ BOOL) ⇛ t

cond′ s t t′ ⩽ (s ⩽ BOOL) ⇛ t′

cond′ s t t′ ⩽ (s ⩽ INT ) ⇛ ERROR

cond′ s t t′ ⩽ (s ⩽ ERROR) ⇛ ERROR

These constraints state that cond′ s t t′ is a lower bound for
each of the terms on the right-hand sides. Thus, we can ob-
tain the optimal implementation of cond′ by simply defining
it to be the greatest such lower bound:

cond′ :: Type → Type → Type → Type
cond′ s t t′ = ((s ⩽ BOOL) ⇛ t)

⊓ ((s ⩽ BOOL) ⇛ t′)
⊓ ((s ⩽ INT ) ⇛ ERROR)
⊓ ((s ⩽ ERROR) ⇛ ERROR)

Taken together, this definition expresses that if the first ar-
gument of a conditional could have type BOOL, then the
type of the result could be that of either branch, while if the
type of the first argument could be INT or a type error, then
the result could be a type error. Note that the component
constraints are not disjoint, i.e. it is possible that more than
one may apply, in which case the results are combined by
taking their greatest lower bound.
By following a similar transformation process, we can

calculate the following set of constraints for add′ that are
equivalent to those developed in Section 7:

add′ t t′ ⩽ (t ⩽ INT ∧ t′ ⩽ INT ) ⇛ INT

add′ t t′ ⩽ (t ⩽ BOOL ∨ t′ ⩽ BOOL) ⇛ ERROR

In turn, we can immediately obtain the optimal definition of
add′ under these constraints:

add′ :: Type → Type → Type
add′ t t′ = ((t ⩽ INT ∧ t′ ⩽ INT ) ⇛ INT )

⊓ ((t ⩽ BOOL ∨ t′ ⩽ BOOL) ⇛ ERROR)

That is, if both arguments could have type INT then the
result could have type INT , while if either argument could
have type BOOL then the result could be a type error.

By using ⇛ and ⊓ as building blocks for cond′ and add′,
we ensure these definitions are monotonic, as we shall con-
sider shortly. In addition, ⇛ and ⊓ satisfy equational laws,
which we can use to simplify the definitions:

(b ⇛ t) ⊓ (b ⇛ t′) = b ⇛ (t ⊓ t′) (⇛ -collect)
(b ⇛ t) ⊓ (b′ ⇛ t′) = b ⇛ t if b′ ⇒ b and t ⩽ t′

(⇛ -subsume)
(b ⇛ t) ⊓ (b′ ⇛ t) = (b ∨ b′) ⇛ t (⇛ -disjunct)

For example, using the ⇛ -collect law, we can simplify the
first half of the definition of cond′:

((s ⩽ BOOL) ⇛ t) ⊓ ((s ⩽ BOOL) ⇛ t′)
= (s ⩽ BOOL) ⇛ t ⊓ t′

In turn, using ⇛ -subsume we can simplify the second half:

((s ⩽ INT ) ⇛ ERROR) ⊓ ((s ⩽ ERROR) ⇛ ERROR)
= (s ⩽ INT ) ⇛ ERROR

Combining these two simplifications, we obtain the follow-
ing equivalent definition of cond′:

cond′ s t t′ = ((s ⩽ BOOL) ⇛ t ⊓ t′)
⊓ ((s ⩽ INT ) ⇛ ERROR)

By straightforward case analyses we can show that the defi-
nitions of cond′ and add′ calculated in this section are equiv-
alent to those in Section 7. The benefit of obtaining the
definitions by directly composing constraints in this manner
is that the process is entirely systematic, and the resulting
definitions are by construction optimal. Moreover, as we shall
see in the next section, this approach will make it easier to
deal with a richer language of types.

We have claimed above that the compositional definitions
of cond′ and add′ are monotonic by construction. To be more
precise, any operation f 𝑥1 · · · 𝑥𝑛 = t on types is monotonic
if the result type t is a constant type such BOOL or INT ,
one of the argument types 𝑥𝑖 , or a combination of these
formed using ⊓ and b ⇛ , where the condition b is in turn
a combination of comparisons 𝑥𝑖 ⩽ c of arguments 𝑥𝑖 with
constant types c using conjunction and disjunction.

For example, cond′ uses the arguments t and t′ along with
the constant type ERROR and combines these using only ⊓
and ⇛ . Moreover, the conditions used by ⇛ only compare
the argument s with constant types. A similar observation
can be made for the definition of add′.

9 Exception Handling
As a more sophisticated example of our approach, we now
extend the language of conditional expressions with support
for throwing and catching an exception:

data Expr = Val Value | Add Expr Expr |
If Expr Expr Expr | Catch Expr Expr

data Value = I Int | B Bool | Throw | Error
Informally, Throw represents an exception that has been
thrown, while Catch x y behaves as the expression x unless
evaluation of x results in an exception being thrown, in
which case the catch behaves as the handler expression y.

To define the semantics for this extended language as an
evaluation function, we first extend the fold operator with a
new parameter to deal with Catch:

folde val add cond catch = f where
f (Val v) = val v

f (Add x y) = add (f x) (f y)
f (If x y z) = cond (f x) (f y) (f z)
f (Catch x y) = catch (f x) (f y)

Using this operator, we can then define an evaluation seman-
tics for expressions by supplying the appropriate operation
for each form of expression:

eval :: Expr → Value
eval = folde id add cond catch

For values, we use the identity function id as previously.
Addition still requires two integers to succeed, but now also
propagates an exception thrown in either argument:

add :: Value → Value → Value
add (I n) (I m) = I (n +m)
add Throw = Throw
add (I ) Throw = Throw
add = Error

Note that the third clause isn’t simply add Throw = Throw,
as this would mean that add Error Throw = Throw, whereas
under the normal left-to-right evaluation order for addition
we expect an error in the first argument to be propagated.
Conditionals are treated as previously, except that an excep-
tion thrown in the first argument is now propagated:

cond :: Value → Value → Value → Value
cond (B b) v w = if b then v else w
cond Throw = Throw
cond = Error

Finally, the new operation catch handles an exception thrown
in the first argument by returning the second, and otherwise
simply returns the value of the first argument:

catch :: Value → Value → Value
catch Throw v = v
catch v = v

We now seek to define a type checker for our language of
exceptional expressions. We begin by extending our previous
language of types to include a type for an exception that has
been thrown, and similarly extend the definition for the
function tval that returns the type of a value:

data Type = INT | BOOL | THROW | ERROR | TOP
tval :: Value → Type
tval (I ) = INT
tval (B ) = BOOL
tval Throw = THROW
tval Error = ERROR

The partial ordering on types is extended as shown below:

TOP

INT THROW BOOL

ERROR

As in Section 8, we include TOP as the greatest element
purely as a technical device to calculate the definition of the
type checker in a compositional manner. Our goal now is to
define a function texp :: Expr → Type that returns the type
of an expression, with the desired behaviour being captured
in the same manner as previously:

texp e ⩽ tval (eval e)
To calculate texp, we use the algebraic approach from Sec-
tion 6, and assume it is defined by folding suitable operations
on types, which themselves remain to be defined:

texp = folde val′ add′ cond′ catch′

The extended fold operator also has an extended fusion prop-
erty, which requires that catch′ is monotonic and satisfies
a homomorphism property. Hence, by fusion, to establish
texp e ⩽ tval (eval e) it suffices to show that the operations
add′, cond′ and catch′ are monotonic, and the function tval
satisfies the following homomorphism properties:

tval (id x) ⩾ val′ x

tval (add x y) ⩾ add′ (tval x) (tval y)
tval (cond x y z) ⩾ cond′ (tval x) (tval y) (tval z)
tval (catch x y) ⩾ catch′ (tval x) (tval y)

We now aim to calculate definitions for the operations on
types val′, add′, cond′ and catch′ that satisfy the above prop-
erties. For values, the property val′ x ⩽ tval (id x) simplifies
immediately to val′ x ⩽ tval x, and hence the optimal defi-
nition for val′ is to simply define:

val′ :: Value → Type
val′ x = tval x

For addition, we proceed by case analysis on the argument
values x and y. When both are integers, the calculation is
the same as previously, resulting in the constraint

add′ t t′ ⩽ (t ⩽ INT ∧ t′ ⩽ INT ) ⇛ INT

The remaining cases lead to further constraints:

Case: x = Throw

add′ (tval Throw) (tval y) ⩽ tval (add Throw y)
⇔ { applying add, tval }
add′ THROW (tval y) ⩽ THROW

⇐ { generalising to arbitrary type }
add′ THROW t′ ⩽ THROW

⇔ { monotonicity of add′ }

add′ t t′ ⩽ THROW , if t ⩽ THROW
⇔ { definition of ⇛ }
add′ t t′ ⩽ (t ⩽ THROW ) ⇛ THROW

Case: x = I n and y = Throw

add′ (tval (I n)) (tval Throw) ⩽ tval (add (I n) Throw)
⇔ { applying add, tval }
add′ INT THROW ⩽ tval Throw

⇔ { applying tval }
add′ INT THROW ⩽ THROW

⇔ { monotonicity of add′ }
add′ t t′ ⩽ THROW , if t ⩽ INT and t′ ⩽ THROW

⇔ { definition of ⇛ }
add′ t t′ ⩽ (t ⩽ INT ∧ t′ ⩽ THROW ) ⇛ THROW

Case: x = Error or x = B b or
(x = I n and (y = Error or y = B b))

add′ (tval x) (tval y) ⩽ tval (add x y)
⇔ { applying add, tval }
add′ (tval x) (tval y) ⩽ ERROR

⇐ { generalising to arbitrary types }
add′ t t′ ⩽ ERROR, if t = ERROR or t = BOOL or

(t = INT and (t′ = ERROR or t′ = BOOL))
⇔ { monotonicity of add′ }
add′ t t′ ⩽ ERROR, if t ⩽ ERROR or t ⩽ BOOL or

(t ⩽ INT and (t′ ⩽ ERROR or t′ ⩽ BOOL))
⇔ { simplifying, ERROR ⩽ BOOL }
add′ t t′ ⩽ ERROR,

if t ⩽ BOOL or (t ⩽ INT and t′ ⩽ BOOL)
⇔ { definition of ⇛ }
add′ t t′ ⩽

(t ⩽ BOOL ∨ (t ⩽ INT ∧ t′ ⩽ BOOL)) ⇛ ERROR

The final case above covers the remaining argument possi-
bilities not covered by the other cases. It is formulated in
a positive manner by explicitly enumerating the remaining
possibilities. This ensures that the calculated definition of
add′ is monotonic by construction.

In conclusion, the above reasoning shows that the follow-
ing four constraints for add′ are together sufficient to satisfy
the homomorphism property for addition:

add′ t t′ ⩽ (t ⩽ INT ∧ t′ ⩽ INT ) ⇛ INT
add′ t t′ ⩽ (t ⩽ THROW ) ⇛ THROW
add′ t t′ ⩽ (t ⩽ INT ∧ t′ ⩽ THROW ) ⇛ THROW
add′ t t′ ⩽

(t ⩽ BOOL ∨ (t ⩽ INT ∧ t′ ⩽ BOOL)) ⇛ ERROR

The optimal solution to these constraints is then obtained
by defining the add′ operation as the greatest lower bound
of each of the right-hand sides:

add′ t t′ =
((t ⩽ INT ∧ t′ ⩽ INT ) ⇛ INT )

⊓ ((t ⩽ THROW ) ⇛ THROW )
⊓ ((t ⩽ INT ∧ t′ ⩽ THROW ) ⇛ THROW )
⊓ ((t ⩽ BOOL ∨ (t ⩽ INT ∧ t′ ⩽ BOOL)) ⇛ ERROR)

Using the ⇛ -disjunct law, we can combine the second and
third component to give a single constraint that returns the
type THROW , resulting in the following final definition:

add′ :: Type → Type → Type
add′ t t′ =

((t ⩽ INT ∧ t′ ⩽ INT ) ⇛ INT )
⊓ ((t ⩽ THROW ∨

(t ⩽ INT ∧ t′ ⩽ THROW )) ⇛ THROW )
⊓ ((t ⩽ BOOL ∨ (t ⩽ INT ∧ t′ ⩽ BOOL)) ⇛ ERROR)

For conditionals, there is only one new case to consider, for
which the calculation proceeds as follows:

Case: x = Throw

cond′ (tval Throw) (tval y) (tval z) ⩽
tval (cond Throw y z)

⇔ { applying cond, tval }
cond′ THROW (tval y) (tval z) ⩽ THROW

⇐ { generalising to arbitrary types }
cond′ THROW t t′ ⩽ THROW

⇔ { monotonicity of cond′ }
cond′ s t t′ ⩽ THROW , if s ⩽ THROW

⇔ { definition of ⇛ }
cond′ s t t′ ⩽ (s ⩽ THROW ) ⇛ THROW

Together with the constraints already calculated in Section 8,
we thus obtain the following constraints for cond′, which
together ensure the required homomorphism property:

cond′ s t t′ ⩽ (s ⩽ BOOL) ⇛ t

cond′ s t t′ ⩽ (s ⩽ BOOL) ⇛ t′

cond′ s t t′ ⩽ (s ⩽ INT ) ⇛ ERROR

cond′ s t t′ ⩽ (s ⩽ ERROR) ⇛ ERROR

cond′ s t t′ ⩽ (s ⩽ THROW ) ⇛ THROW

We can then immediately obtain the optimal definition of
cond′ that satisfies these constraints. As in Section 8, we can
then simplify the resulting definition by combining the first
and second component as well as the third and fourth, to
give the following definition:

cond′ :: Type → Type → Type → Type
cond′ s t t′ = ((s ⩽ BOOL) ⇛ t ⊓ t′)

⊓ ((s ⩽ INT ) ⇛ ERROR)
⊓ ((s ⩽ THROW ) ⇛ THROW )

Finally, for the catch operation that handles exceptions there
are two cases to consider, namely whether the first argument
results in an exception being thrown or not:

Case: x = Throw

catch′ (tval Throw) (tval y) ⩽ tval (catch Throw y)
⇔ { applying catch, tval }
catch′ THROW (tval y) ⩽ tval y

⇐ { generalising to arbitrary type }
catch′ THROW t′ ⩽ t′

⇔ { monotonicity of catch′ }
catch′ t t′ ⩽ t′, if t ⩽ THROW

⇔ { definition of ⇛ }
catch′ t t′ ⩽ (t ⩽ THROW ) ⇛ t′

Case: x = Error or x = B b or x = I n

catch′ (tval x) (tval y) ⩽ tval (catch x y)
⇔ { applying catch }
catch′ (tval x) (tval y) ⩽ tval x

⇐ { generalising to arbitrary types }
catch′ t t′ ⩽ t, if t = ERROR or t = BOOL or t = INT

⇔ { monotonicity of catch′ }
catch′ t t′ ⩽ t′′,

if t ⩽ t′′ and t′′ ∈ {ERROR, BOOL, INT }
⇔ { definition of ⇛ }
catch′ t t′ ⩽ (t ⩽ t′′) ⇛ t′′,

if t′′ ∈ {ERROR, BOOL, INT }

The above reasoning shows that the following four con-
straints for catch′ are together sufficient to satisfy the homo-
morphism property for catching exceptions:

catch′ t t′ ⩽ (t ⩽ THROW ) ⇛ t′

catch′ t t′ ⩽ (t ⩽ ERROR) ⇛ ERROR

catch′ t t′ ⩽ (t ⩽ BOOL) ⇛ BOOL

catch′ t t′ ⩽ (t ⩽ INT ) ⇛ INT

From this we obtain the optimal definition for catch′:

catch′ :: Type → Type → Type
catch′ t t′ = ((t ⩽ THROW ) ⇛ t′)

⊓ ((t ⩽ ERROR) ⇛ ERROR)
⊓ ((t ⩽ BOOL) ⇛ BOOL)
⊓ ((t ⩽ INT ) ⇛ INT )

In summary, using the techniques developed in previous
sections, we have calculated a type checker for the language
of exceptional expressions. Moreover, each of the derived
operations on types add′, cond′ and catch′ is built using the
pattern described in Section 8 and is hence monotonic by
construction, so there is no need to manually check this.

10 Checked Exceptions
The type checker developed in the previous section is cor-
rect, as it satisfies the specification texp e ⩽ tval (eval e),
but is less informative than we might wish. In this section

we explain the problem, and show how it can be solved by
further exploiting algebraic properties.

The problem is how exceptions are treated in conditionals.
In particular, if we have a conditional where one branch
produces a regular value and the other branch throws an ex-
ception, this will be regarded as being ill-typed. For example,
if we represent the conditional expression

if True then 1 else Throw

in our language, then applying the type-checking function
texp will give the result ERROR, meaning that the expres-
sion contains a type error. This behaviour arises from the
definition of the conditional operation on types:

cond′ s t t′ = ((s ⩽ BOOL) ⇛ t ⊓ t′)
⊓ ((s ⩽ INT ) ⇛ ERROR)
⊓ ((s ⩽ THROW ) ⇛ THROW )

Using this definition, the example expression has type

cond′ BOOL INT THROW = INT ⊓ THROW = ERROR

In contrast, in Haskell the similar expression

if True then 1 else throw SomeException

type checks fine as an integer. In Haskell, this is achieved
throw having a polymorphic type, Exception e ⇒ e → a.
However, this approach might be viewed as unsatisfactory,
as the polymorphic result type gives no indication that an
exception may be thrown. It would be preferable to solve the
problem by making the type system more informative.

To deal with exceptions more informatively, the type sys-
tem needs to be revised so that INT ⊓ THROW is not simply
ERROR. This can be achieved by adding a new type INT?
that represents a value that could be either an integer or an
exception, and a new type BOOL? that represents a value
that could be either a logical value or an exception:

data Type = INT | INT? | BOOL | BOOL?
| THROW | ERROR | TOP

The desired behaviour of the new types is specified by
INT ⊓ THROW = INT? and BOOL ⊓ THROW = BOOL?,
which can be realised by refining the ordering on types to:

TOP

INT THROW BOOL

INT? BOOL?

ERROR

In this manner, we now have an extra level of types between
the error type and the original value types, which allow us
to represent values that could potentially be exceptions.

We now consider what further changes are required. First
of all, because the type of values is unchanged, the function
tval :: Value → Type does not require any modification.
In turn, calculations of the constraints for the type oper-

ations val′, add′, cond′ and catch′ also remain unchanged,
because the calculations only use positive facts about the
ordering, and any ordering on types t ⩽ t′ in the original
definition of Type still holds in the extended definition.
Finally, because we defined val′, add′, cond′ and catch′

as optimal solutions for the calculated constraints, these
definitions also satisfy the same constraints for the extended
type, and are still optimal. Hence, the definition of the type
checker texp :: Expr → Type can also remain unchanged.

In summary, due to the manner in which the type checker
was constructed in the previous section, no changes are
required beyond modifying the language of types. That is,
the existing definitions can be used without modification
with the extended form of types. Crucially, however, the type
checker can now give more informative results. For example,
type checking if True then 1 else Throw now gives:

cond′ BOOL INT THROW = INT ⊓ THROW = INT?

That is, the type checker now indicates that the expres-
sion could either produce an integer or raise an exception,
whereas previously it simply gave a type error. As another
example, consider catch (if True then 1 else Throw) 2.
As shown above, the conditional expression has type INT?,
hence the whole expression has type:

catch′ INT? INT = INT ⊓ INT = INT

In this manner, the type checker recognises that catch han-
dles the possibility of an exception being raised in the condi-
tional, ensuring that an integer is always returned.

11 Related Work
The idea of calculating type checkers from (in)equational
specifications is inspired by the compiler calculation method-
ology of Bahr and Hutton [2, 3]. However, calculating type
checkers has required further refinements: using an ordered
version of fold fusion to calculate inequational constraints,
and a means to obtain optimal solutions for these constraints
in a compositional manner.

A type checker is a special case of the notion of an abstract
interpreter [5]. The soundness of an abstract interpreter 𝑓𝑎
with respect to a concrete interpreter 𝑓𝑐 is typically formu-
lated as a Galois connection, which consists of two inequali-
ties, 𝑓𝑎 ◦𝛼 ⩽ 𝛼 ◦ 𝑓𝑐 and 𝛾 ◦ 𝑓𝑎 ⩽ 𝑓𝑐 ◦𝛾 , for suitable abstraction
and concretisation functions 𝛼 and 𝛾 . In our setting, the con-
crete interpreter 𝑓𝑐 is eval, the abstract interpreter 𝑓𝑎 is the
type checker texp, and the abstraction function 𝛼 is tval. Our
methodology dispenses with both the concretisation func-
tion 𝛾 and the second inequality. While we only consider
simple languages here, there is reason for optimism that
such a simplification may generalise to higher-order and

non-terminating languages. In particular, Bahr and Hutton
[2] have shown that the compiler calculation technique of
Meijer [11] based on adjunctions, which generalise Galois
connections, can be simplified to a single equation by trans-
forming the higher-order semantics of the source language
into first-order form by defunctionalisation.
Our calculation of a type checker for a language with

exceptions naturally led to a type system with checked ex-
ceptions [7, 9], a feature popularised in Java [8]. Exceptions
and exception handling are a special case of algebraic effects
and effect handling [15]. Type and effect systems have been
proposed to check for the presence and proper handling of
effects [4]. This suggests further work to explore how type
and effects systems for algebraic effects can be calculated
using the kind of techniques presented in this article.

12 Conclusion and Further Work
We showed how the calculational approach to program con-
struction can be applied to the design of type-checking func-
tions. In particular, we identified algebraic properties of both
the evaluation semantics and the type language that allowed
us to derive correct-by-construction type checkers in a prin-
cipled and compositional manner.

There are a number of interesting topics for further work
in this area. First of all, it would be useful to be able to cal-
culate the language of types at the same time as the type
checker itself, in a similar manner to how Bahr and Hut-
ton [2] calculate the language of code at the same time as a
compiler. Secondly, it is important to consider how the ap-
proach can be scaled to more sophisticated source language
and type system features, such as non-termination, higher-
order languages, computational effects, type constructors,
and polymorphism. And finally, it would be interesting to
calculate typing rules as opposed to type checkers, and we
already have some promising results in this direction.

